

HDL Lab Report

Group 03

Course Semester: SoSe 2023

Supervisor: Viktor Weinelt

Department: Integrierte Elektronische Systeme (IES)

Prof. Dr.-Ing. Klaus Hofmann

HDL Lab Report by

Xue Wenxuan, 2500838

Jingang Zhang, 2705413


Abdenassir El Amin, 2773160

Darmstadt



TECHNISCHE
UNIVERSITÄT
DARMSTADT

Electrical Engineering and
Information Technology
Department



Contents

1	Introduction	4
2	Implementation of ADC	5
2.1	Discussion and design	5
2.2	Implementation of Sample & Hold	5
2.3	Implementation of Quantizer	5
2.4	Testbench and Waveform	6
3	Implementation of Controller	7
3.1	Discussion and design	7
3.2	Binary search algorithm	7
3.2.1	Principle and Prerequisites	7
3.2.2	First binary search	8
3.2.3	Index of Vppg_max and Vppg_min	8
3.2.4	Second binary search	8
3.3	Implementation of state machine	8
3.3.1	INITIAL	8
3.3.2	FIND_SETTING	8
3.3.3	IDLE	8
3.3.4	FIRST_DC_COMP	9
3.3.5	SWEEP_DC_COMP	9
3.3.6	INCREASE PGA GAIN	9
3.3.7	OSCILLATE	10
3.4	Testbench and Waveform	12
4	Implementation of FIR-Filter	13
4.1	Discussion and Design	13
4.2	Implementation of Verilog code	13
4.3	Testbench and Waveform	13
5	Top Module of Controller & FIR-Filter	16
5.1	Discussion and Design	16
5.2	Testbench and Waveform	16
6	Synopsis Design Version and SDF Back Annotation	17
6.1	Synopsis Design Vision	17
6.2	SDF Back Annotation	17
7	Layout of Top Module	19

7.1	Discussion and Design	19
8	Summary	20
8.1	Final Result	20
8.2	The issues we resolved	20
8.2.1	Sample frequency	20
8.2.2	Blocking vs Non-Blocking	21
8.3	Outlook and Challenges	21
8.3.1	Frequency problem	21
8.3.2	Burr phenomena	21
8.3.3	DC_Drive issues	21
9	Appendix	22
9.1	ADC converter	22
9.2	Controller	23
9.3	FIR-Filter	28
9.4	Top-module of Controller & FIR-Filter	31

1 Introduction

The process to accelerate digital designs in Integrated Circuits (ICs) heavily relies on high-level Hardware Description Language (HDL) such as Verilog or VHSIC(Very High-Speed Integrated Circuits) Hardware Description Language (VHDL). In this lab, we used HDL step by step to realize the digital design part of a pulse oximetry method. This involved creating three main components: an Analog-to-Digital Converter (ADC), a controller, and a Finite Impulse Response-Filter (FIR-Filter).

To begin, we used Verilog-A to build an ADC capable of converting analog signals into 8-bit digital signals. Subsequently, a testbench was created in Cadence Virtuoso[1] to verify its operational accuracy.

Following this, Verilog code was written to separately implement the controller and FIR-Filter. The controller's function involves adjusting the values of DC Compensation and increasing the Programmable-gain amplifier (PGA) Gain to control the inputs of the analog frontend. It produces two separate 8-bit streams for the red LED and the infrared LED outputs. The role of the FIR Filter is to remove high-frequency noise present in the system. For both of these components, testbenches were written using ModelSim[2] to ensure logical correctness. Further validation and algorithm optimization were performed by creating and utilizing testbenches in Cadence Virtuoso to observe waveform signals. After optimizing the algorithm, we enabled the controller to determine appropriate DC Compensation and PGA Gain values for the LEDs within 3 seconds.

Following that, Synopsis[3] was used to validate and modify the code for both components until they were successfully synthesized.

The synthesized code was then subjected to functional logic verification using a testbench in Cadence Virtuoso. A Top Module was developed to instantiate and interconnect the controller and two FIR-Filters, resulting in an integrated block. This block was likewise synthesized and verified.

However, merely simulating the block for ideal timing isn't enough to guarantee the correct operation of the gate-level circuit. Hence, Standard Delay Format (SDF) was employed to further simulate the block under realistic timing conditions. Finally, Cadence Innovus and related scripts were utilized to position all input and output pins and optimize the size of chip area. This resulted in the final design layout of the block.

This report will treat each component as a separate chapter, and within each chapter we will discuss its functions, design, simulations, verifications, and various aspects.

2 Implementation of ADC

2.1 Discussion and design

As referenced from the CAD4SOC Lab Exercise 2 manual, the ideal ADC converter consists of two important components: the Sample&Hold circuit and the Quantizer. The former is used to sample the the analog input signal every millisecond, while the latter converts the continuous analog signal values sampled and held in the holding circuit into discrete digital values and outputs them. The sampling frequency for this ADC converter design is $1kHz$, and its resolution is 8 bits. The following sections will discuss and show the code implementation and observed waveform results for these two components.

2.2 Implementation of Sample & Hold

According to the working principle of the Sample & Hold circuit, we need to capture the current input analog signal value on every rising edge of the clock signal and store it. This stored value is then passed to the Quantizer until the next sampling event occurs. The specific code implementation is shown in Appendix.

2.3 Implementation of Quantizer

Based on the design requirements to quantize continuous analog input signals into an 8-bit digital output signal, it is determined that the number of quantization levels required is $2^8 = 256$. The output value *result* is obtained using the formula 2.1:

$$result = 256 \times \frac{V_{in} - v_{RefL}}{v_{RefH} - v_{RefL}} + 0.5 \quad (2.1)$$

where v_{RefL} represents the lower reference voltage of $0V$ and v_{RefH} is the upper reference voltage of $1.8V$. Addition of 0.5 ensures that the result is rounded to the nearest quantization level. Additionally, to prevent overflow and underflow, the result is bounded within the range of 0 to 255. Then, the integer result does AND operation with each bit. If the result is 1, it means that the corresponding output bit is set to a high voltage of $1.8V$ (logic 1), and vice versa. The code is shown in Appendix. A significant portion of our code was based on the code discussed in the CAD4SOC Lab Exercise 2.

2.4 Testbench and Waveform

In the process of creating a testbench for the ADC, we utilize the analog output signal *Vppg* generated by *Frontend_TB* as the input to the ADC. By observing the waveforms of *Vppg* and the generated 8-bit digital output signals *D<7:0>* in below figure 2.2, we can confidently determine that the ADC is functioning properly.




Figure 2.1: Testbench for ADC




Figure 2.2: Waveform of input and outputs in ADC

3 Implementation of Controller

3.1 Discussion and design

The function of this controller is to find the corresponding DC compensation and maximum PGA gain for the red LED and infrared LED as soon as possible and accurately, so that in the case of these two LEDs being turned on respectively, the average value of the analog signal V_{ppg} with PGA gain of 0 will be close to 0.9V and with this maximum PGA gain the signal will not be clipped. Then turn on both LEDs alternately for 10ms when the corresponding settings are found for the red LED and the infrared LED respectively. At the same time, an 8-bit digital signal is output for the red LED and the infrared LED respectively.

The following sections discusses the algorithm used to implement the controller, the use of state machines, and the waveform simulated by testbench.

3.2 Binary search algorithm

3.2.1 Principle and Prerequisites

Binary search is an efficient algorithm to find a valid DC_Comp, adhering to a logarithmic time complexity of $O(\log(n))$. To use this algorithm in this implementation of controller we must first check the monotony between the DC_Comp and the value of V_{ppg} . Because binary search can only be used to deal with sorted list of items, which means the need of a continuous monotone function. By adjusting the value of DC_Comp in `Initial_Values.v` and examining the waveform of V_{ppg} in `Frontend_TB`, it can be observed that for both LEDs, as DC_Comp is increased, the value of V_{ppg} exhibits a monotonic decrease. The results of checking are shown in the following table 3.1.

Table 3.1: Monotony between DC_Comp and V_{ppg}

LED	DC_Comp	Waveform of V_{ppg}
red/infrared	0	line near 1.75v
infrared	64	wave near 0.9v
red	104	wave near 0.9v
red/infrared	127	line near 0v

3.2.2 First binary search

In the beginning of our program we will do a rough binary search of DC_Comp using the first sample value of Vppg. This step is to make sure that there is an obvious wave in order to help us to find the index of Vppg_max and Vppg_min in one cycle.

3.2.3 Index of Vppg_max and Vppg_min

After performing 1000 samples on the waveform of Vppg over one complete cycle, we obtained the index values corresponding to Vppg_max and Vppg_min within that cycle. Initially, we used the method of summing and averaging 1000 samples to determine if Vppg is close to 0.9 V. However, this method is slow (1000 samples per cycle). Since the waveform of Vppg is similar to sinus signal, we later used to sample only 1 cycle to get the index value of Vppg_max and the index value of Vppg_min. With these two index values, we can get the wave's Vppg_max, Vppg_min, and Vppg_median for each subsequent cycle. By selecting Vppg_median, we can determine whether Vppg is close to 0.9v or not.

3.2.4 Second binary search

After calculating the index of Vppg_median, we will sample the Vppg value around this index 15 times, and compare it with 0.9 V. Finally we can get a relatively accurate valid DC_Comp value by binary search again.

3.3 Implementation of state machine

3.3.1 INITIAL

Upon reception of an asynchronous reset, the system will initiate by entering the INITIAL state, where it will appropriately assign initial values to each register. In this we will assign the nextState to FIND_SETTING as well as almost assigning every register the value 0. This is important because it allows us to have a logic understanding of what our current values are and in case of errors, which happened a lot during synthesizing, we could debug it.

3.3.2 FIND_SETTING

Upon receiving the Find_Setting signal, we will start go into IDLE state and initiate the process to find the DC_Comp and PGA_Gain settings for both LEDs.

3.3.3 IDLE

This state was implemented to set the initial settings for the red LED and infrared LED before finding the correct settings for DC_Comp and PGA_Gain. We assign the register DC_left the value 0 and the register DC_right the value of our parameter MITTEL_VOLTAGE, which happens to be 127, so we would have these values before we would enter our first binary search in order to find our DC compensation(DC_Comp) value.

3.3.4 FIRST_DC_COMP

Upon finishing the initial settings for the red LED and infrared LED, we did the first DC compensation sweep. The aim of this first initial DC compensation sweep was to generate a wave of Vppg instead of a line so we could extract the index of Vppg_max and Vppg_min. For the generation of a wave we use binary search algorithm. How the algorithm works was described in the section Binary search algorithm.

3.3.5 SWEEP_DC_COMP

In Sweep DC Comp we use binary search to find the value nearest to 0.9 V or 127 as digital value. In the first step we calculate the median index value as

```
1 min_index + max_index >>1;
```

and we will at most need 15 clock frequencies to finish the binary search. In the next step we check if the current digital voltage from the Fingerclip is lower or higher than MITTEL_VOLTAGE, which is 0.9 V as an analog value or as a digital value 127. If its smaller than 127, then we save the old DC Comp value inside the register DC_right, which was found in the previous state FIRST_DC_Comp and calculate the new DC_Comp value by taking the middle value in between the previous DC_Comp and DC_left value as.

```
1 DC_Comp + DC_left >>1;
```

Otherwise, if the value is higher than 0.9 V, then we would save the old DC_Comp value in the register DC_left and calculate the new DC_Comp as.

```
1 DC_Comp + DC_right >>1;
```

After 15 clock frequencies or if Vppg is equal to 127, we have found an optimal value for the DC compensation and we assign the register nextState the state INCREASE_PGA_GAIN and shift into the next state INCREASE_PGA_GAIN.

3.3.6 INCREASE PGA GAIN

Once we have found our DC comp settings we start to adjust the PGA Gain. The idea on how we approach to increase the PGA Gain is by finding the signals index for its maximum value and increase the PGA Gain every clock cycle, as the signals amplitude isn't changing that much within 15 clock periods. In the first statement

```
1 (counter >= max_index) && counter <= (max_index + 15);
```

we check if our counter is higher than the maximum index and equal or less than 15. If that is the case and if Vppg < MAX_RAND_VOLTAGE, then we increase the PGA Gain by one step. Otherwise, we already have found our maximum PGA Gain and temporarily save it in gaintemp_max

```
1 gaintemp_max <= PGA_Gain;
```

If the counter reaches 15, then we reset the PGA Gain value for the next PGA Gain calculation and start storing the PGA Gain and DC Comp settings for the red and infrared LED respectively.

```
1 PGA_Gain <= 4'd0;  
2 start <= 1;
```

Now we check whether the red LED(LED_RED) or the infrared LED(LED_IR) is on. If the red LED is on, then we would save the DC_Comp value and gaintemp_max in registers as our DC comp and PGA gain settings. We would also reset gaintemp_max value to 0 and assign nextState to IDLE, as we have to find a DC compensation and PGA gain value for the infrared LED.

```
1 DC_Comp_RED <= DC_Comp;  
2 PGA_Gain_RED <= gaintemp_max;  
3 nextState <= IDLE;
```

Otherwise if infrared LED is on, we would reset the same settings and save the DC compensation value and PGA gain value in registers as well as change the nextState to Oscillate.

```
1 DC_Comp_IR <= DC_Comp;  
2 PGA_Gain_IR <= gaintemp_max;  
3 nextState <= OSCILLATE;
```

3.3.7 OSCILLATE

Once we have established both settings, they will be saved in registers and we will start to oscillate between the two LED's with a frequency of 100 Hz. We managed to create the oscillation between the two LED's with a frequency of 100 Hz by having a counter that goes up to 10 which divides the controller frequency of 1 kHz by 10. We'll provide more detailed explanations of how the code operates in appendix.





Figure 3.1: State machine of Controller


3.4 Testbench and Waveform

For the controller's testbench, we mainly used the given Frontend_HDL_Lab_TB to directly connect our designed ideal ADC and controller to the modules LED_Drive_Gen and FRONTEND_with_supply. By changing the value of Drive_Strength between 8 to 12 in LED_Drive_Gen, we get the waveform shown in below Figure 3.2.


By observing the waveforms, we can see that when Drive_Strength is 10, the controller we designed can accurately find the DC_Comp and PGA_Gain for both LEDs within 3s. However, when Drive_Strength is changed, even though the value of DC_Comp is found correctly, the Vppg is chipped due to the inaccurate PGA_Gain. Unfortunately, due to time constraints, we weren't able to find the optimization for this problem.




(a) Waveform of Controller with Drive_Strength:= 8



(b) Waveform of Controller with Drive_Strength:= 10



(c) Waveform of Controller with Drive_Strength:= 11



(d) Waveform of Controller with Drive_Strength:= 12

Figure 3.2: Different Drive_Strength for controller

4 Implementation of FIR-Filter

4.1 Discussion and Design

In order to reduce high-frequency noise in the system, two identical low-pass FIR-Filters with 8-bit digital signal inputs from controllers for red LED and infrared LED have to be implemented. Since FIR-Filter is purely a digital circuit component, it can be developed entirely using Verilog instead of Verilog-A. According to research, the working principle of a FIR-Filter involves multiplying the input digital signal $x[n]$ with a set of coefficients (b_i), and then summing the products to obtain the output signal $y[n]$. The formula 4.1[4] is as follows:

$$y[n] = b_0x[n] + b_1x[n-1] + \cdots + b_Nx[n-N] = \sum_{i=0}^N b_i \cdot x[n-i] \quad (4.1)$$


For this design, the FIR-Filter's sampling frequency is $500Hz$, and with the given 22 coefficients, the output results in a 20-bit filtered digital signal output. We provided specific code implementation section in Appendix.

4.2 Implementation of Verilog code


From the equation 4.1, we can see that to build the FIR-Filter, we need 22 shift registers acting as buffer. Each of these can hold an 8-bit input signal. Whenever the clock signal rises, the `input_data<7:0>` gets stored in `shift_regs[0]`, and the prior content of `shift_regs[0]` moves to `shift_regs[1]`, and so forth. Due to the symmetrical nature of the FIR-Filter coefficients, we also need 11 add registers. For instance, `add_regs[0]` holds the sum of values from `shift_regs[0]` and `shift_regs[21]`, while `add_regs[1]` holds the sum of values from `shift_regs[1]` and `shift_regs[20]`, and so on. Afterward, 11 order registers `order_regs` are used to keep the product of multiplying each add register's value by the corresponding coefficients. Finally, each of the products is added together, resulting in the value that becomes the 20-bit output. This algorithm utilizes only 11 multipliers instead of 22, which helps to save chip area probably. In Appendix shows the specific implementation code block.

4.3 Testbench and Waveform

To simulate and verify the correctness of this code logic, we initially created a testbench in ModelSim. By observing the following figure4.1, we noted that when the input was set to 1, the output was the sum of all coefficients (decimal: 1386) or (hexadecimal: 056A). Similarly, when the input was set to 255, the output was the product of 255 and the sum of coefficients (decimal: 353430) or (hexadecimal: 56496). This observation provides evidence that the logic is correct.




(a) The waveform of the input == 1




(b) The waveform of the input == 255

Figure 4.1: Comparison of the waveform between the input := 1 and input := 255


Similarly, a testbench is created in Cadence Virtuoso, and a sinus signal with frequencies of $1Hz$, $10Hz$, $20Hz$, and $100Hz$ and an amplitude of $900mv$ is used as the analog input signal, V_{ppg} , which is converted by an ideal ADC to an 8-bit digital signal, $D<7:0>$, as the input signal to the FIR-Filter. As shown in following figure 4.2, as the input signal of the FIR-Filter, the amplitude of the output signal decreases to 0 as the frequency increases, thus realizing the characteristic of passing low frequency and blocking high frequency.




(a) Waveform of input frequency: $1Hz$



(b) Waveform of input frequency: $10Hz$



(c) Waveform of input frequency: $20Hz$



(d) Waveform of input frequency: $100Hz$

Figure 4.2: Different input frequencies of inputs in FIR

5 Top Module of Controller & FIR-Filter

5.1 Discussion and Design

After designing the controller and the FIR-Filter, we need to combine these two sub-modules by Verilog code to form the final top module. This top module should include a controller and two identical FIR-Filters, each filtering the output signals of the red LED and the infrared LED, respectively. The implementation code is shown in the Appendix.

5.2 Testbench and Waveform

After putting the symbol generated from the code of this top module into the given Frontend_HDL_Lab_TB, the waveform in the figure 5.1 below is obtained by using the testbench simulation. Observing the waveform, we can see that the top module works normally with a value of 10 for Drive_Strength.




Figure 5.1: Waveform of top module with Drive_Strength:= 10

6 Synopsis Design Version and SDF Back Annotation

6.1 Synopsis Design Vision

Synopsis Design Vision was the chosen logic synthesis tool for the conversion of HDL designs into gate-level HDL netlists. The process began with the synthesis of the Filter and Controller components. These components were translated from high-level descriptions into gate-level representations suitable for hardware implementation.

For our Clock, we chose a period of 1 ms. We have faced several problems by synthesis and therefore we adapted our code towards making it synthesized. First it is about initial value. Neither initial command or assigning initial values during declaration can not be synthesised. So we must add an initial state in our state machine. We remove every task and function and write them out in the respective state. Additionally, we removed all blocking assignments and transferred them into non-blocking assignments and encountered many errors when using more than one always block.

Following successful individual synthesis, the Controller and FIR_Filter were combined into a single Top Module. The outcome of these efforts was summarized in following Table 6.1. This table held results, documenting aspects such as the area of each module, power consumption, and so on.

Table 6.1: The reports of each module

module	Area/ μm^2	power consumption/nW	slack time/ns
controller	5487.68	88.35	999987.50
FIR-Filter	11372.87	227.88	999991.56
controller & FIR-Filter	28311.07	538.35	999987.69

6.2 SDF Back Annotation

After we had synthesized all of the modules, we proceeded to perform a gate-level simulation using SDF Back Annotation. The SDF Back Annotation involve simulations of the synthesized Filter and Controller. This step ensured accurate performance by considering realistic timing and standards. The waveforms of the AMS Simulation for the controller testbench, FIR-Filter testbench and the top module testbench were displayed in following Figure 6.1. During the AMS Simulation of our controller testbench and FIR-Filter testbench, we observed the intentional presence of noise in the digital output stream. Unfortunately, we did not observe the 20-bit noise-filtered output in the top module, but only the input signal Vppg. We guessed that the FIR-Filter

in the top module was not working properly due to not holding realistic time constraints and we were not able to solve this problem unfortunately.






Figure 6.1: Waveform of each module with AMS Simulation

7 Layout of Top Module

7.1 Discussion and Design

First we set all the pins equably on the both sides of chip, which is different with frequency in our design. Second it is common to leave a blank space in the floorplan where no standard cell exists. it is not possible to have every standard cell close to each other in the floorplan, as this would lead to routing congestion problems and would not be conducive to timing. So we adjust the size ratio, which we choose 0.7, because in our design the value higher than it will cause shortcuts. This is to minimize the chip area while ensuring that the circuit works correctly.




Figure 7.1: Layout of top module with clock frequency 1kHz

8 Summary

8.1 Final Result

After we finished all necessary tests, we collect important information of parameters of our design, that shows the quality of our design.

Table 8.1: Achieved results

Frequency	1 kHz	100 MHz	Max.Frequency(1 GHz)
time to find valid setting in controller	3 s	X	X
Area Synthesis	28311.07 μm^2	28292.58 μm^2	34187.89 μm^2
Area Layout	23685.2 μm^2	23715.960 μm^2	29308 μm^2
Power consumption in Synthesis	538.35 nW	98988 nW	1.4744 mW
Power consumption in Layout	474.5 nW	501.7 nW	3853 nW

As the result shows, we achieve a very fast find-setting time, but our design consumes a little bit larger area and power consumption.

Table 8.2: Filename table

Exercise	Path and Modulename	Path and Testbenchname
Verilog-A ADC	HDL_LAB_Gruppe03/ADC	HDL_LAB_Gruppe03/ADC_TB
Verilog Controller	src/Controller.v/Controller	HDL_LAB_Gruppe03/Controller_TB
Verilog Filter	src/FIR_Filter.v/FIR_Filter	HDL_LAB_Gruppe03/FIR_Filter_TB
Verilog Controller&FIR	src/Controller_FIR.v/Controller_FIR	HDL_LAB_Gruppe03/Controller_FIR_TB

8.2 The issues we resolved

8.2.1 Sample frequency

In the beginning of the creation of the ADC, we got no digital signal at all. There are lots of high frequency part in the output. Then we realize, that the Nyquist Sampling Theorem is violated, where the sampling frequency is more than twice the highest frequency of the signal in order for the signal to be recovered without distortion.

8.2.2 Blocking vs Non-Blocking

In ModelSim we use blocking assignment in the second part of state machine as the script shows. But during the simulation in Cadence we found, that blocking assignment can not give a correct output because of timing sequence. So we decided to change all assignments to non-blocking. Meanwhile we have to change the conditional statements to original value minus one because every value will always be assigned after the end of a block.

8.3 Outlook and Challenges

During our first SDF Back Annotation we separately do the simulation for controller and the filter, which works well. But when we combine them together, the find-setting part still works properly but the oscillation part appeared to have lots of cut-off. We will discuss the possible reasons in following section.

8.3.1 Frequency problem

Due to the nature of FIR Filters, the input signal at 100 Hz will be converted into max value. We think the reason of that problem might be, that when red and infrared oscillate with 100Hz, the frequency of input signal increases and is eliminated by FIR.

8.3.2 Burr phenomena

Second when signals in digital circuits go through the logic unit connection, there must be a delay. In addition, the signal of the high to low level conversion also requires a certain transition time, due to the existence of these two factors, the level of multi-channel signal value changes, in the instant of the signal changes, the output of the combined logic has a sequential order, not the same time changes, often appear some incorrect spike signal, which are competing adventures, that will generate burrs.

8.3.3 DC_Drive issues

The controller we designed is fast and correctly executes when DC_Drive is 10, but once its value is changed to 8 or 12, Vppg is clipped. This means we have to optimize the code of INCREASE_PGA_Gain state, towards making it not leading to clipping once the DC_Drive value is changed to 8 or 12.

9 Appendix

9.1 ADC converter

```
1 // VerilogA for HDL_LAB_Gruppe03, SimpleHold4ADC, veriloga
2
3 /**
4 This module is a sample and hold block for ADC
5 **/
6 'include "constants.vams"
7 'include "disciplines.vams"
8
9 module SimpleHold4ADC(AGND, clk, vIn, vOut);
10
11     // four inout ports: AGND, clk, vIn, and vOut.
12     inout AGND;
13     electrical AGND; // declared as electrical, indicating electrical nature.
14     inout clk;
15     electrical clk;
16     inout vIn;
17     electrical vIn;
18     inout vOut;
19     electrical vOut;
20
21     // the clock threshold voltage is 0.9 V
22     parameter real clkThreshold = 0.9 from [0:inf];
23     real vOutTemp;
24
25     analog begin
26         // waits for a rising edge of the clock (clk) that crosses the clock threshold voltage
27         @(cross(V(clk,AGND) - clkThreshold,1))begin
28             vOutTemp = V(vIn,AGND); // rising edge is detected, the input voltage vIn stored
29             in vOutTemp.
30             end
31
32             V(vOut,AGND) <+ vOutTemp; // vOutTemp is assigned to the output voltage vOut
33
34     end
35 endmodule
```

Verilog-A code block of Sample & Hold circuit

```
1 // VerilogA for HDL_LAB_Gruppe03, Quantizer4ADC, veriloga
2 /**
3 This module is quantizer for analog-to-digital converter (ADC)
4 **/
```

```

5 'include "constants.vams"
6 'include "disciplines.vams"
7
8 module Quantizer4ADC(AGND, D, vIn);
9
10     // three inout ports: AGND, D, and vIn.
11     inout AGND;
12     electrical AGND; //declared as electrical, indicating electrical nature
13     inout[7:0] D;
14     electrical[7:0] D;
15     inout vIn;
16     electrical vIn;
17
18     // end of automatically generated header
19     // the logic levels are 0 to 1.8. V
20     parameter real vLow = 0 from [0:inf];
21     parameter real vHigh = 1.8 from [0:inf];
22
23     // the reference voltage is 1.8. V
24     parameter real vRefL = 0 from [0:inf];
25     parameter real vRefH = 1.8 from [0:inf];
26
27     // the number of quantization levels
28     localparam integer levels = 256;
29
30     integer result;
31
32 analog begin
33     // the output value is calculated arithmetically
34     result = levels*(V(vIn,AGND) - vRefL)/(vRefH - vRefL) + 0.5;
35
36     // overflow and underflow are caught here
37     if (result > levels-1)
38         result = levels-1;
39     else if (result < 0)
40         result = 0;
41
42     // the integer result is converted into bits
43     V(D[0]) <+ (result & 1) ? vHigh : vLow;
44     V(D[1]) <+ (result & 2) ? vHigh : vLow;
45     V(D[2]) <+ (result & 4) ? vHigh : vLow;
46     V(D[3]) <+ (result & 8) ? vHigh : vLow;
47     V(D[4]) <+ (result & 16) ? vHigh : vLow;
48     V(D[5]) <+ (result & 32) ? vHigh : vLow;
49     V(D[6]) <+ (result & 64) ? vHigh : vLow;
50     V(D[7]) <+ (result & 128) ? vHigh : vLow;
51 end
52
53 // end of module
54 endmodule

```

Verilog-A code block of Quantizer

9.2 Controller

```

1 INITIAL: begin
2     nextState      <= FIND_SETTING;
3     counter        <= 4'd0;
4     PGA_Gain_RED   <= 0;
5     PGA_Gain_IR    <= 0;
6     DC_Comp_temp   <= 0;
7     DC_Comp_IR     <= 0;
8     DC_Comp_RED    <= 0;
9     red_on_flag    <= 1;
10    temp_min       <= 255;
11    temp_max       <= 0;
12    min_index      <= 0;
13    max_index      <= 0;
14    LED_IR         <= 0;
15    LED_RED        <= 1;
16    PGA_Gain       <= 4'd0;
17    gaintemp       <= 15;
18    LED_Drive      <= 8;
19    CLK_Filter     <= 0;
20    gaintemp_max   <= 0;
21    gaintemp_min   <= 0;
22    start <= 0;
23 end

```

Verilog code block of INITIAL state

```

1 FIND_SETTING: begin
2     if(Find_Setting) begin
3         nextState <= IDLE;
4     end
5 end

```

Verilog code block of FIND SETTING state

```

1 IDLE: begin
2     // reset
3     DC_left  <= 0;
4     DC_right <= MITTEL_VOLTAGE;
5     case(red_on_flag) //Here we check whether we have do DC_Sweep with RED LED or IR LED.
6     0: begin //Here we do 1 LED_IR DC_Sweep in order to get a waveform so we can determine
7         min and max index of a wave.
8         nextState <= FIRST_DC_COMP;
9         LED_IR    <= 1;
10        LED_RED   <= 0;
11        DC_Comp   <= 0;
12        temp_min  <= 255;
13        temp_max  <= 0;
14    end
15    1: begin //Here we do 1 DC_Sweep for LED_RED in order to get a waveform so we can
16        determine min and max index of a wave.
17        nextState <= FIRST_DC_COMP;
18        LED_IR    <= 0;
19        LED_RED   <= 1;
20        DC_Comp   <= 0;
21        temp_min  <= 255;
22        temp_max  <= 0;
23    end

```



```

22  2: begin //In previous states we have calculated min and max index and here we do
multiple DC_Comp changes around the median value until it matches 0.9 Voltage.
23
24      if (Vppg < temp_min && LED_RED == 1) begin
25          temp_min <= Vppg;
26          min_index <= counter; //index_counter
27          nextState <= IDLE;
28      end
29      if (Vppg > temp_max && LED_RED == 1) begin
30          temp_max <= Vppg;
31          max_index <= counter; //index_counter
32          nextState <= IDLE;
33      end
34
35      if (counter == 999 && LED_RED == 1)begin //index_counter
36          nextState <= SWEEP_DC_COMP; // set led red on
37      end
38      if(LED_IR == 1) begin
39          nextState <= SWEEP_DC_COMP;
40      end
41  end
42
43  default;;
44  endcase
45  end

```

Verilog code block of IDLE state

```

1  FIRST_DC_COMP:begin
2      if ((DC_right - DC_left) > 1)begin
3          if(Vppg < MITTEL_VOLTAGE)begin
4              DC_right <= DC_Comp;
5              DC_Comp_temp <= DC_Comp + DC_left;
6              DC_Comp <= DC_Comp_temp >>1;
7          end else if(Vppg > MITTEL_VOLTAGE)begin
8              DC_left <= DC_Comp;
9              DC_Comp_temp <= DC_Comp + DC_right;
10             DC_Comp <= DC_Comp_temp >> 1;
11         end else begin
12             DC_left <= DC_Comp;
13             DC_right <= DC_Comp;
14         end
15     end else begin
16         red_on_flag <= 2;
17         nextState <= IDLE;
18     end
19 end

```

Verilog code block of FIRST DC COMP state

```

1  SWEEP_DC_COMP: begin
2      if (counter < (((min_index +max_index) >> 1) + 7) && counter >= (((min_index + max_index)
>> 1) - 7))begin
3          if(Vppg < MITTEL_VOLTAGE)begin
4              DC_right <= DC_Comp;
5              DC_Comp_temp <= DC_Comp + DC_left;
6              DC_Comp <= DC_Comp_temp >>1;
7

```

```

8      end else if(Vppg > MITTEL_VOLTAGE)begin
9          DC_left      <= DC_Comp;
10         DC_Comp_temp  <= DC_Comp + DC_right;
11         DC_Comp       <= DC_Comp_temp >> 1;
12
13     end else begin
14         DC_left      <= DC_Comp;
15         DC_right     <= DC_Comp;
16         nextState    <= INCREASE_PGA_GAIN;
17     end
18 end else if (counter >= (((min_index +max_index) >> 1) + 7)) begin
19     nextState      <= INCREASE_PGA_GAIN;
20 end
21 end

```

Verilog code block of SWEEP DC COMP state

```

1 INCREASE_PGA_GAIN: begin
2     if((counter >= max_index) && counter <= (max_index + 4'd15)) begin
3         if(Vppg < MAX_RAND_VOLTAGE) begin
4             PGA_Gain <= PGA_Gain + 4'd1;
5         end
6         else begin
7             gaintemp_max <= PGA_Gain; //Once it would start to clip the PGA_Gain will be
            saved in a register.
8         end
9         if(counter == (max_index + 4'd15)) begin //Here we make sure to store the biggest
            PGA_Gain value and reset the PGA_Gain value once we go through all 15 PGA_Gains
10             PGA_Gain <= 4'd0;
11             start <= 1;
12         end
13     end
14     if(start) begin
15         start <= 0;
16         if(2000 < gaintemp_max) begin
17             red_on_flag <= 0 ;
18             counter <= 0; //reset
19
20             if(LED_IR) begin // store actual DC_Comp setting of LED IR in
                register DC_Comp_IR
21                 DC_Comp_IR <= DC_Comp; //store the PGA_Gain setting of LED IR in
                register PGA_Gain_IR
22                 PGA_Gain_IR <= gaintemp_min; //start switching between LED RED UND LED IR
                with a frequency of 100 Hz
23                 nextState <= OSCILLATE;
24                 PGA_Gain <= 4'd0; //reset
25                 gaintemp <= 15;
26                 gaintemp_max <= 0;
27                 gaintemp_min <= 0;
28             end
29
30             if(LED_RED) begin
31                 DC_Comp_RED <= DC_Comp; // store actual DC_Comp setting of LED RED in
                register DC_Comp_RED
32                 PGA_Gain_RED <= gaintemp_min; //store the PGA_Gain setting of LED RED in
                register PGA_Gain_RED
33                 nextState <= IDLE; //We have to start with LED_RED and switch to
                LED IR

```

```

34         PGA_Gain      <= 4'd0;           //reset of PGA_Gain
35         gaintemp      <= 15;
36         gaintemp_max   <= 0;
37         gaintemp_min   <= 0;
38         LED_RED        <= 0;
39         LED_IR         <= 1;
40     end
41
42     end else begin
43         red_on_flag     <= 0 ;
44         counter         <= 0; //reset
45
46         if(LED_IR) begin
47             DC_Comp_IR  <= DC_Comp;       // store actual DC_Comp setting of LED IR
48             in register DC_Comp_IR
49             PGA_Gain_IR  <= gaintemp_max;  //store the PGA_Gain setting of LED IR in
50             register PGA_Gain_IR
51             nextState    <= OSCILLATE;     //start switching between LED RED UND LED
52             IR with a frequency of 100 Hz
53             PGA_Gain     <= 4'd0; //reset
54             gaintemp     <= 15;
55             gaintemp_max <= 0;
56             gaintemp_min <= 0;
57             LED_RED      <= 1;
58             LED_IR       <= 0;
59         end
60
61         if(LED_RED) begin
62             DC_Comp_RED  <= DC_Comp;       // store actual DC_Comp setting of LED RED
63             in register DC_Comp_RED
64             PGA_Gain_RED  <= gaintemp_max;  //store the PGA_Gain setting of LED RED in
65             register PGA_Gain_RED
66             nextState    <= IDLE;          //We have to start with LED_RED and switch
67             to LED IR
68             PGA_Gain     <= 4'd0;          //reset of PGA_Gain
69             gaintemp     <= 15;
70             gaintemp_max <= 0;
71             gaintemp_min <= 0;
72             LED_RED      <= 0;
73             LED_IR       <= 1;
74         end
75     end
76 end
77 end
78 end

```

Verilog code block of INCREASE PGA GAIN state

```

1 OSCILLATE: begin
2     CLK_Filter <=~CLK_Filter; //Create Clock for FIR_Fitler
3     if(counter == 9) begin //We count up to 10 to create periods of 10 ms
4         LED_RED <= ~LED_RED;
5         LED_IR  <= ~LED_IR;
6         counter <= 0; //reset
7
8         // set dc and pga parameter for next LED RED
9         if(LED_IR) begin
10             DC_Comp      <= DC_Comp_RED;
11             PGA_Gain     <= PGA_Gain_RED;

```

```

12         IR_ADC_Value <= Vppg; //output
13     end
14
15     // set dc and pga parameter for next LED IR
16     if(LED_RED) begin
17         DC_Comp <= DC_Comp_IR;
18         PGA_Gain <= PGA_Gain_IR;
19         RED_ADC_Value <= Vppg; //output
20     end
21
22 end
23 end

```

Verilog code block of OSCILLATE state

9.3 FIR-Filter

```

1  /**
2  This module is FIR Filter, that can remove high frequency noise in system
3  **/
4  module FIR_Filter(
5      input wire CLK_Filter,          // Clock for the filter
6      input wire [7:0] input_data,    // Input data from ADC (0 to 255)
7      input wire rst_n,               // Reset signal (active-low)
8      output reg [19:0] output_data   // Filtered output (20-bit)
9  );
10
11  parameter NUM_SHI_REGS = 22;       // Number of shift registers that as buffer
12  parameter NUM_ADD_REGS = 11;       // Number of add registers
13
14  reg [7:0] shift_regs [NUM_SHI_REGS-1:0]; // 22 shift registers as buffer to store
15  // input data
16  reg [8:0] add_regs [NUM_ADD_REGS-1:0]; // to store the summed data of first and
17  // last shift register
18  // Each order multiplies with the according coefficient
19  reg [16:0] order[NUM_ADD_REGS-1:0];
20  wire [7:0] COEFFICIENTS [NUM_ADD_REGS-1:0]; // to store FIR Filter coefficients value
21
22  // 11 FIR filter coefficients
23  // Only 11 multipliers needed because of the symmetry of FIR filter coefficients
24  assign COEFFICIENTS[0] = 8'd2;
25  assign COEFFICIENTS[1] = 8'd10;
26  assign COEFFICIENTS[2] = 8'd16;
27  assign COEFFICIENTS[3] = 8'd28;
28  assign COEFFICIENTS[4] = 8'd43;
29  assign COEFFICIENTS[5] = 8'd60;
30  assign COEFFICIENTS[6] = 8'd78;
31  assign COEFFICIENTS[7] = 8'd95;
32  assign COEFFICIENTS[8] = 8'd111;
33  assign COEFFICIENTS[9] = 8'd122;
34  assign COEFFICIENTS[10] = 8'd128;
35
36  always @(posedge CLK_Filter or negedge rst_n) begin
37      if (!rst_n) begin
38          // reset the value in all registers to 0

```

```

37     shift_regs[0]   <= 8'b0;
38     shift_regs[1]   <= 8'b0;
39     shift_regs[2]   <= 8'b0;
40     shift_regs[3]   <= 8'b0;
41     shift_regs[4]   <= 8'b0;
42     shift_regs[5]   <= 8'b0;
43     shift_regs[6]   <= 8'b0;
44     shift_regs[7]   <= 8'b0;
45     shift_regs[8]   <= 8'b0;
46     shift_regs[9]   <= 8'b0;
47     shift_regs[10]  <= 8'b0;
48     shift_regs[11]  <= 8'b0;
49     shift_regs[12]  <= 8'b0;
50     shift_regs[13]  <= 8'b0;
51     shift_regs[14]  <= 8'b0;
52     shift_regs[15]  <= 8'b0;
53     shift_regs[16]  <= 8'b0;
54     shift_regs[17]  <= 8'b0;
55     shift_regs[18]  <= 8'b0;
56     shift_regs[19]  <= 8'b0;
57     shift_regs[20]  <= 8'b0;
58     shift_regs[21]  <= 8'b0;
59
60     add_regs[0]     <= 9'b0;
61     add_regs[1]     <= 9'b0;
62     add_regs[2]     <= 9'b0;
63     add_regs[3]     <= 9'b0;
64     add_regs[4]     <= 9'b0;
65     add_regs[5]     <= 9'b0;
66     add_regs[6]     <= 9'b0;
67     add_regs[7]     <= 9'b0;
68     add_regs[8]     <= 9'b0;
69     add_regs[9]     <= 9'b0;
70     add_regs[10]    <= 9'b0;
71
72     order[0]        <= 17'b0;
73     order[1]        <= 17'b0;
74     order[2]        <= 17'b0;
75     order[3]        <= 17'b0;
76     order[4]        <= 17'b0;
77     order[5]        <= 17'b0;
78     order[6]        <= 17'b0;
79     order[7]        <= 17'b0;
80     order[8]        <= 17'b0;
81     order[9]        <= 17'b0;
82     order[10]       <= 17'b0;
83
84     output_data <= 20'd0;
85
86 end else begin
87     // all shift registers to store input data
88     shift_regs[0]   <= input_data;
89     shift_regs[1]   <= shift_regs[0];
90     shift_regs[2]   <= shift_regs[1];
91     shift_regs[3]   <= shift_regs[2];
92     shift_regs[4]   <= shift_regs[3];
93     shift_regs[5]   <= shift_regs[4];
94     shift_regs[6]   <= shift_regs[5];

```

```

95     shift_regs[7]   <= shift_regs[6];
96     shift_regs[8]   <= shift_regs[7];
97     shift_regs[9]   <= shift_regs[8];
98     shift_regs[10]  <= shift_regs[9];
99     shift_regs[11]  <= shift_regs[10];
100    shift_regs[12]  <= shift_regs[11];
101    shift_regs[13]  <= shift_regs[12];
102    shift_regs[14]  <= shift_regs[13];
103    shift_regs[15]  <= shift_regs[14];
104    shift_regs[16]  <= shift_regs[15];
105    shift_regs[17]  <= shift_regs[16];
106    shift_regs[18]  <= shift_regs[17];
107    shift_regs[19]  <= shift_regs[18];
108    shift_regs[20]  <= shift_regs[19];
109    shift_regs[21]  <= shift_regs[20];
110
111    // register data are summed first and last
112    add_regs[0]      <= shift_regs[0] + shift_regs[21];
113    add_regs[1]      <= shift_regs[1] + shift_regs[20];
114    add_regs[2]      <= shift_regs[2] + shift_regs[19];
115    add_regs[3]      <= shift_regs[3] + shift_regs[18];
116    add_regs[4]      <= shift_regs[4] + shift_regs[17];
117    add_regs[5]      <= shift_regs[5] + shift_regs[16];
118    add_regs[6]      <= shift_regs[6] + shift_regs[15];
119    add_regs[7]      <= shift_regs[7] + shift_regs[14];
120    add_regs[8]      <= shift_regs[8] + shift_regs[13];
121    add_regs[9]      <= shift_regs[9] + shift_regs[12];
122    add_regs[10]     <= shift_regs[10] + shift_regs[11];
123
124    // Each order multiplies with the according coefficient
125    order[0]         <= COEFFICIENTS[0] * add_regs[0];
126    order[1]         <= COEFFICIENTS[1] * add_regs[1];
127    order[2]         <= COEFFICIENTS[2] * add_regs[2];
128    order[3]         <= COEFFICIENTS[3] * add_regs[3];
129    order[4]         <= COEFFICIENTS[4] * add_regs[4];
130    order[5]         <= COEFFICIENTS[5] * add_regs[5];
131    order[6]         <= COEFFICIENTS[6] * add_regs[6];
132    order[7]         <= COEFFICIENTS[7] * add_regs[7];
133    order[8]         <= COEFFICIENTS[8] * add_regs[8];
134    order[9]         <= COEFFICIENTS[9] * add_regs[9];
135    order[10]        <= COEFFICIENTS[10] * add_regs[10];
136
137    // output
138    output_data <=  order[0] + order[1] + order[2] + order[3]
139                   + order[4] + order[5] + order[6] + order[7]
140                   + order[8] + order[9] + order[10];
141
142 end
143 endmodule

```

Verilog code block of FIR-Filter

```

1  /**
2  This FIR_Filter Testbench is used to simulate Waveform of outputs of FIR in ModelSim
3  **/
4  `timescale 1ms/1us
5  module FIR_Filter_TB;
6

```

```

7  reg      CLK_Filter;    //The clock signal of Filter
8  reg [7:0] input_data;
9  wire [19:0] output_data;
10 reg      rst_n;        //An asynchronous positive edge reset signal
11
12 //Instantiate FIR Filter Module
13 FIR_Filter DUT (
14     .CLK_Filter(CLK_Filter),
15     .input_data(input_data),
16     .output_data(output_data),
17     .rst_n(rst_n)
18 );
19
20 initial begin
21     CLK_Filter = 0;
22     input_data = 0;
23
24     // Reset the filter
25     #5 rst_n = 0;
26     #5 rst_n = 1;
27
28     // Testbench stimulus
29     #10 input_data = 1;    // Test input data
30
31     // #10 input_data = 255; // Test input data
32     #100 $stop;           // End simulation after some time
33 end
34
35 always begin
36     #1 CLK_Filter = ~CLK_Filter; // Toggle clock every 1 time units
37 end
38
39 endmodule

```

Verilog code block of FIR-Filter testbench

9.4 Top-module of Controller & FIR-Filter

```

1  /**
2  This module is the top module of controller and FIR filter
3  **/
4  module Controller_FIR (
5      input  [7:0]  VPPG,          //Binary output Signal of the ADC
6      input        Find_Setting,   //Reset to start find Settings
7      input        CLK,           //The clock signal
8      input        rst_n,         //An asynchronous positive edge reset signal
9
10     output [6:0]  DC_Comp,        //Used to change the DC_Offset
11     output [3:0]  pga_Gain,       //Used to increase the Gain
12     output        LED_IR,        //Controls Infrared LED
13     output        LED_RED,       //Controls RED LED
14     output [19:0] Out_RED_Filtered, //The filtered Bit stream output of RED LED
15     output [19:0] Out_IR_Filtered, //The filtered Bit stream output of RED IR
16     output [3:0]  LED_Drive       //Sets the LED brightness
17 );

```

```

18
19 // the wires to connect Contollor module and 2 FIR Filters moudule
20 wire [7:0] IR_ADC_Value, RED_ADC_Value;
21 wire      CLK_Filter;
22
23 //Instantiate Controller_Model
24 Controller ct( .Vppg(VPPG),
25               .Find_Setting(Find_Setting),
26               .CLK(CLK),
27               .rst_n(rst_n),
28               .DC_Comp(DC_Comp),
29               .PGA_Gain(pga_Gain),
30               .CLK_Filter(CLK_Filter),
31               .LED_IR(LED_IR),
32               .LED_RED(LED_RED),
33               .IR_ADC_Value(IR_ADC_Value),
34               .RED_ADC_Value(RED_ADC_Value),
35               .LED_Drive(LED_Drive));
36
37 //Instantiate the FIR_Filter Module for RED_ADC_Value
38 FIR_Filter red( .CLK_Filter(CLK_Filter),
39               .input_data(RED_ADC_Value),
40               .rst_n(rst_n),
41               .output_data(Out_RED_Filtered));
42
43 //Instantiate the FIR_Filter Module for IR_ADC_Value
44 FIR_Filter ir( .CLK_Filter(CLK_Filter),
45               .input_data(IR_ADC_Value),
46               .rst_n(rst_n),
47               .output_data(Out_IR_Filtered));
48
49 endmodule

```

top_module

List of Abbreviation

ICs Integrated Circuits

HDL Hardware Description Language

VHDL VHSIC(Very High-Speed Integrated Circuits) Hardware Description Language

ADC Analog-to-Digital Converter

FIR-Filter Finite Impulse Response-Filter

PGA Programmable-gain amplifier

SDF Standard Delay Format

Bibliography

- [1] *Cadence Virtuoso*. <https://www.cadence.com/>.
- [2] *ModelSim*. <https://eda.sw.siemens.com/en-US/ic/modelsim/>.
- [3] *Synopsys*. <https://www.synopsys.com/>.
- [4] *FIR-Filter equation*. https://en.wikipedia.org/wiki/Finite_impulse_response.

List of Figures

2.1	Testbench for ADC	6
2.2	Waveform of input and outputs in ADC	6
3.1	State machine of Controller	11
3.2	Different Drive_Strength for controller	12
4.1	Comparison of the waveform between the input := 1 and input := 255	14
4.2	Different input frequencies of inputs in FIR	15
5.1	Waveform of top module with Drive_Strength:= 10	16
6.1	Waveform of each module with AMS Simulation	18
7.1	Layout of top module with clock frequency 1kHz	19

List of Tables

3.1	Monotony between DC_Comp and Vppg	7
6.1	The reports of each module	17

8.1	Achieved results	20
8.2	Filename table	20